

Customizing SAS Graphs Using the Annotate Facility and Global Statements

Abstract

SAS/GRAPH is a powerful, highly customizable graphics package that, with knowledge of a few tools, can produce most types of graphs. SAS/GRAPH PROCedures provide canned methods for creating graphics; global statements control the general appearance of the graphics, and the ANNOTATE facility can customize, or even produce, most anything that the procedures can produce. This paper will lay the foundation of the functionality of global statements, heavily emphasize the capabilities of the ANNOTATE facility, and suggest additional resources to use when customizing graphics.

Assumptions

To use the annotate facility and the global statements, certain prerequisites must be met:

- A working knowledge of basic SAS/GRAPH PROCedures.
- A working knowledge of how SAS/GRAPH can be made to display or print graphics on your operating system.
- A thorough understanding of the DATA step will be very important, since the ANNOTATE facility relies on the creation of a dataset to operate.
- An understanding of SAS/BASE global statements (e.g. TITLE, FOOTNOTE, OPTIONS, etc.). Also, knowing that global statements execute immediately upon reaching those statements in the code is important.

Global Statements

Global statements, such as TITLE and FOOTNOTE, are used in SAS/BASE to enhance output. SAS/GRAPH offers a number of graphic-specific global statements, plus options on the TITLE and FOOTNOTE statements, to customize graphics procedures.

A number of text options can be used in either the GOPTIONS statement (to change the default text options), or in any global statement to change the next text displayed by that statement. Keep in mind that the actual syntax in the global statement might be slightly different - double check the SAS Reference manual if not sure. Some of these options are found in table 1.

Option Syntax	Description and Example
Colors = (list of colors)	defines the colors available to SAS/GRAPH. Example: goptions colors=(red green blue);
Ftext = text-font	defines the default font for all graphics text. Example: goptions ftext=SWISS;
Htext = text height <units>	defines the default height for all text. Example: goptions htext=.5 in
Hby = by-line height <units>	defines the default height for by line text. Example: goptions hby=1 cm

Table 1 Examples of options available to many SAS/GRAPH global statements.

In addition, it is helpful to be able to reset the graphics options/global statements. To do this, you can issue the statement:

```
goptions reset=ALL|GLOBAL|<statement name>;
```

This will reset ALL graphics options and global statements, just the GLOBAL statements, or the global statement that you specified.

Some common global statements can be found in Table 2.

Global Statement	Common Options	Description	Example
SYMBOLn (often used for PROC GPLOT).	Interpol=join Value=DOT	joins each point with a line. puts a graphics DOT at each point.	Symbol I=join value=DOT colors=(red);
PATTERNn (often used for PROC GCHART).	Value=X3	Defines the pattern displayed in the chart. See Figure 15.1 on pages 366-367 of the SAS/GRAPH Reference Manual	Pattern value=x3 colors=(red green);
AXISn	Label=('axis') Minor=none	Defines the label of the axis. Specifies what the 'minor' tick marks will look like. NOTE: in a PROC that uses AXIS, the PROC must be told which axis to use. For example, in PROC GPLOT, the options for the PLOT statement should include: haxis=axis1 vaxis=axis2 (linking the Horizontal and Vertical axes, respectively).	Axis1 label=('Quarter') minor=none;

Table 2 Some common SAS/GRAPH global statements.

For further definition and details, see chapters 9-17 of the SAS/GRAPH Reference manual.

CAUTION: In GOPTIONS, you can define the colors available or leave as the default for your system. If more than one color is available, global statements such as SYMBOL will use every color before using the next SYMBOL statement. For example, take the following code:

```
goptions colors=(red green blue);
symbol i=join value=STAR;
symbol2 i=join value=DOT;
```

On the graph, the first line will be graphed in red with stars at each point. The second line will be in green with stars at each point, the third in blue with stars. Then, the fourth line will be red with dots, etc. This caution applies to other statements as well, such as PATTERN.

The ANNOTATE Facility

The annotate facility is a fancy term which, simply put, means a set of step by step instructions for labeling or drawing (i.e. annotating) on a graph. The set of instructions is stored in a specially set-up

dataset, called an annotate dataset. The SAS/GRAPH PROCedure then specifies the name of the annotate dataset, usually by specifying ANNOTATE=<dataset> as an option on the PROCedure line.

TIP: The annotate facility is only useful when a graph needs to be updated later. For one-time only customizations, use the interactive graphics editor, by simply typing EDIT at the command line of the GRAPH window.

The ANNOTATE Dataset

The annotate dataset is made up of many variables which can be broken out into three areas of functionality:

- What to do.
- How to do it.
- Where to do it.

The first record on this dataset gives the first instruction. The second record gives the second instruction, and so forth. The instructions, in plain English, would look similar to those found in Table 3.

Dataset Name:	What to do	How to do it	Where to do it
SASUSER.RETAIL			
Step 1	Put a star on the graph.	Using a special font that will give a star.	At the point of interest.
Step 2	Move to a new point on the graph.		Just below and to the right at the previous point.
Step 3	Draw a diagonal line.		Down and to the right.
Step 4	Label the graph 'Opened New Store This Quarter'.		At the end of the diagonal line.

Table 3 Steps, in English, for the example in this paper.

Area of Functionality	Variable Name	Character or Numeric	Length	Comments
What to do	FUNCTION	Character	8	What do you want the observation to do? We will look at 'LABEL'. Some other values are 'DRAW','POLY','POLYCONT','MOVE','SYMBOL'.
	TEXT	Character	1 - 100	Tells what to display, if appropriate.
How to do it	COLOR	Character	8	What color?
	STYLE	Character	1	Selects the font or pattern of the text or line.
	SIZE	Numeric	8	Selects the height or thickness of the text or line.
	HSYS	Character	1	Selects the coordinate system* used by the size variable.
Where to do it	X	Numeric	8	The horizontal coordinate.
	XSYS	Character	1	Selects the coordinate system* used by the x variable.
	Y	Numeric	8	The vertical coordinate.
	YSYS	Character	1	Selects the coordinate system* used by the y variable.
	POSITION	Character	1	Tells where to draw text in relation to the point (x,y). Values are typically '1' to '9' or 'A' to 'F'. For example, position='9' positions the text just below the point, and left aligns it. See the chart on page 522 of the SAS/GRAPH Reference manual for further definition.

* Coordinate System is a fancy term defining the units and origin of SIZE, X, and Y. A value of XSYS=2, for example, indicates that the values of x are the data values and the origin is at the bottom left corner of the axes. See the charts on pages 476-477 of the SAS/GRAPH Reference manual for more definition.

Table 4 Variables which control each area of functionality.

The variables, which control each area of functionality, are found in table 4.

To perform the task above, the following code could be used:

```
*****;
* Create the annotate dataset.;
*****;
data label;
  set sasuser.retail;
  length function $8 text $200;

  color='RED';
  size=1;

  * Sizes based on the cells of the
  * Graphics Ouptut Area.;
  hsys='4';

  * X and Y coordinate systems based on
  * Actual data values.;
  xsys='2'; * X Coordinate S data values;
  ysys='2';

  if date='01jan1969'd then do;

    * Put a STAR on the graph.;
    function='LABEL';
    style='SPECIAL';
    text='M'; * The special character for a star.
    position='5'; * Centered on the point.;
    x = date;
    y = sales;
    output;

    * Move to the coordinates (x+25,y+25);
    function='MOVE';
    x = x + 25;
    y = y - 25;
    output;


    * Draw a line from (x,y) down to the right.;
    function='DRAW';
    position='9'; * One cell below, left-align.;
    x = x + 45;
    y = y - 25;
    output;

    * Put the label onto the graph.;
    function='LABEL';
    position='F'; * 1/2 cell below, left-align.;
    style='SWISS';
    text='Opened New Store This Quarter';
    output;

  end;
run;

symbol i=join;
title height=2 'Net Sales Over Time';
title2 height=1 'As Reported at the Annual
Shareholders Meeting';
footnote justify=left height=.5
'Net sales are defined as Gross Sales minus
all operating expenses.';
axis1 label=('Quarter') minor=none;
axis2 label=('Total Sales');
proc gplot data=sasuser.retail annotate=label;
  plot sales*date / haxis=axis1 vaxis=axis2;
run;
quit;
```

This code produces the following basic graph:



Additional Topics

Several other tools are useful for customizing SAS graphs.

- The annotate macros will assist you in creating an annotate dataset by providing a shorter way of specifying each function than declaring the name of each variable. SAS/GRAPH Software Reference, Volume 1, pages 570-587. TIP: Be sure to use the %annomac macro first in your data step; this macro compiles and makes the other annotate macros available for use.
- The Data Set Graphics Interface (or DSGI for short) is a totally separate facility for creating SAS graphics directly from within a SAS dataset. Information on this tool can be found in the SAS/GRAPH Software Reference, Volume 1, Chapters 20-21. This is a powerful method for creating graphics.
- Interactive graphics editor. Type 'edit' at the command line in the graph window to get started.

References

- SAS/GRAPH Software, Usage, Version 6, First Edition, pages 751-769.
- SAS/GRAPH Software, Examples, Version 6, First Edition.
- SAS/GRAPH Software, Reference, Version 6, First Edition, pages 570-587, chapters 9-17, and chapters 20-21.

Contact

Jeffery D. Gilbert
 Trilogy Consulting Corporation
 5148 Lover's Lane
 Kalamazoo, MI 49002
 (888) 825 3225
JefferyGilbert@hotmail.com